Р. ХУДЯКОВ

Практика работы с периферийным процессором УКНЦ

Прежде чем программа будет работать в ОЗУ ПП, ее нужно туда загрузить. Рассказ об этой операции начнем с массива параметров (далее — МП), формат которого приведен на первой распечатке.

MASSIV: .Byte   0       ;В этом байте будет содержаться код ошибки или 0

COMAND: .Byte   0       ;Команда (01-ВЫДЕЛИТЬ ПАМЯТЬ, 02-ОСВОБОДИТЬ ПАМЯТЬ, 10-ЧТЕНИЕ, 20-ЗАПИСЬ, 30-ПУСК)
TYPDEV: .Word   32      ;Устройство - периферийный процессор

ADR$PP: .Word   0       ;Адрес в ОЗУ ПП

ADR$CP: .Word   0       ;Адрес в ОЗУ ЦП

LENGTH: .Word   0       ;Длина программы в словах (не байтах!)

ADRMAS: .Word   MASSIV  ;Слово всегда содержит начальный адрес массива параметров

        .Word   401             ;Стоповый элемент (используется при передаче)

ZAGR:   .Enable LSB

        MOV     #ADRMAS,R0      ;Адрес МП в R0

        MOV     #5,R1

        BR      1$              ;Передать четыре байта

2$:     MOVB    (R0)+,@#176676  ;(поле ADRMAS и слово, содержащее 401)

1$:     TSTB    @#176674

        BPL     .-4

        SOB     R1,2$

        RTS     PC              ;Вернуться

Команды «Выделить память», «Освободить память», «Чтение», «Запись», «Пуск». Для правильного выполнения команды «Выделить память» поле COMAND МП должно содержать код 01, TYPDEV — код 32, a LENGTH — длину загружаемой программы. Значения остальных полей игнорируются. После пересылки МП в поле ADR$PP ПП возвращает адрес, начиная с которого после команды «ЗАПИСЬ» и будет размещена пересылаемая программа. При повторном запросе выделение памяти производится из оставшейся свободной. Если размер программы превосходит размер свободной памяти в ПП, то в поле LENGTH возвращается длина свободной памяти, а поле MASSIV не будет равно нулю.

Для команды «Освободить память» поля ADR$PP и LENGTH должны содержать те же параметры, которые задавались при выделении данного сегмента, иначе команда не будет выполнена, а в поле MASSIV будет возвращено ненулевое значение.

Для команды «Чтение» («Запись») поле ADR$CP содержит адрес ОЗУ ЦП, откуда (куда) будет пересылаться программа. Для команды «Пуск» ADR$PP содержит стартовый адрес программы. Сама программа должна быть оформлена как подпрограмма (заканчиваться RTS PC) и быть перемещаемой, так как место ее размещения заранее неизвестно.

На распечатке 2 приведена подпрограмма загрузки МП в ПП.

ZAGPP:  .Enable LSB

     MOVB    #1,COMAND       ;Отработать команду "ВЫДЕЛИТЬ ПАМЯТЬ"

     MOV     (R5),OCP        ;Через R5 передается табличка из

     CALL    ZAGR            ;двух слов. 1-е – длина программы, 2-е - ее начальный адрес в ОЗУ ЦП

     TSTB    MASSIV          ;A памяти хватит?

     BEQ     1$              ;Да

     MOV     #NOMEM,R0       ;Выведем сообщение об ошибке

     EMT    351              ;"Не хватает памяти"

     CMP     (R5)+,(R5)+     ;Увеличим R5 на четыре

     SEC                     ;Установим признак ошибки

     BR      2$              ;И на выход

1$:  MOVB    #20, COMAND     ;Отработать "ЗАПИСЬ"

     MOV     (R5)+,LENGTH    ;Из таблички 1-е слово в LENGTH,

     MOV     (R5)+,ADR$CP    ;2-е в ADR$CP

     CALL    ZAGR

     MOVB    #30,COMAND      ;Отработать "ПУСК"

     CALL    ZAGR

     MOVB    #2,COMAND       ;Отработать "ОСВОБОДИТЬ ПАМЯТЬ"
     MOV     -4(R5),ADR$CP
     CALL    ZAGR
2$:  RTS     R5              ;Вернуться

Она передает по каналу К2 адрес массива параметров и два байта завершения передачи, нормализующие состояние канала.

На распечатке 3 приведена полная подпрограмма загрузки программы, ее пуска и последующего освобождения памяти ПП по окончании работы.

.MACRO  $ZAGPP  STADR,ENDADR

        JSR     R5,ZAGPP

        .WORD   <ENDADR-STADR/2>,STADR
.ENDM   $ZAGPP

Формат вызова подпрограммы приведен на распечатке 4.

     .TITLE  MYPROG
START:  $ZAGPP  STARTPP,ENDPP        ;Загрузить программу
     EMT     350                     ;Выход в монитор

;Здесь находятся подпрограммы, сообщения и массив параметров

STARTPP: INC    (PC)+                ;Эту программу можно заменить любой другой

     .WORD  0                        ;---*---*---*---*---*--

     RTS     PC                      ;Вернуться в ЦП

ENDPP:

     .END    START

А вот полностью законченная программа.

↓-----АДРЕС

|0000|0000|0000|0000||0000|0000|0000|0000|

 бел.желт.фиол.красн.бирюз.зел.голуб.черн.

Итак, программа в ПП загружена.

Работа с регистрами внешних устройств не отличается от обычной. Адресное пространство ОЗУ ПП имеет диапазон 0—77777, ПЗУ ПП — 100000—177777, т. е. 32 Кбайт ОЗУ и 32 Кбайт ПЗУ. ПП (как и ЦП) имеет свой пультовый отладчик, облегчающий отладку программ (вход в него осуществляется из меню установок одновременным нажатием клавиш УПР и @). Работать с ним нужно так же, как и с пультовым отладчиком ЦП.

Основное внимание мы уделим дополнительным возможностям, которые могут быть использованы только при помощи ПП. Начнем со средств отображения. Для начала «пощелкаем» палитрой. В УКНЦ любой цвет, выводимый на экран, можно представить любым другим (поменять местами красный и зеленый, фиолетовый и бирюзовый и т. д.). Допустим, вы поменяли синий на белый. Теперь все, что вы запишите в голубой план, на экране появится белым цветом, причем если какая-либо другая программа не изменит значения соответствующей ячейки памяти, то даже после выхода в монитор все, что должно было бы отображаться на экране голубым, будет отображаться белым, включая графическую информацию. Замена цветов происходит очень быстро (практически мгновенно), на этом можно построить интересные эффекты.

Формат ячеек управления аппаратным отображением приведен на рисунке (таких пар слов четыре, их «полномочия» приведены в табл. 1). Каждый из цветов представлен четырьмя битами. Байты каждой из тетрад означают (слева направо):

яркость цвета (1 — 100 % яркости, 0 — 50%);

красный цвет (1 — красный есть, 0 — нет);
зеленый цвет (1 — есть, 0 — нет);
голубой цвет (1 — есть, 0 — нет).
     MOV      #2470,R0        ;Адрес палитры экрана в R0

     MOV      (R0),-(SP)      ;Запомнить в стеке текущее значение

     MOV      2(R0),-(SP)     ;палитры

     CLR      (R0)            ;Выключить экран

     CLR      2(R0)           ;(все цвета на экране представить черным)

     MOV      #100000,@#177010

     MOV      #30.,R1

2$:  MOV      #20.,R2

1$:  MOVB     #377,@#177012   ;Нарисовать квадрат

     INC      @#177010

     SOB      R2,1$

     ADD      #<80.-20.>,@#177010
     SOB      R1,2$

     MOV      #1,R1           ;"Щелкаем" палитру.

3$:  MOV      R1,R3

     .REPT    4               ;Четыре раза ASL R3

     ASL      R3

     .ENDR

     BIC      #360,(R0)

     BIS      R3,(R0)         ;Квадрат  отображается

     CLR      R2              ;цветами от голубого до белого

     SOB      R2,.            ;<-задержка

     SOB      R2,.            ;<-задержка

     INC      R1

     CMP      #8.,R1

     BNE      3$

     MOV      (SP)+,2(R0)     ;Возвращаем предыдущие значения

     MOV      (SP)+,(R0)      ;ячейкам

     RTS      PC              ;Возврат

На машинах литеры 5 завода «Квант» из схемы контроллера цветного монитора удалена ИС 155ЛП9, поэтому яркостный бит не оказывает влияния на изображение; на почти половине машин завода «СЭМЗ» ИС 155ЛП9 заменена на ИС 155ЛН2, поэтому в яркостный бит для получения 100 % яркости нужно записать 0, а для 50 % — 1. Все это справедливо только для цветного монитора, на отображение градаций яркости на черно-белом мониторе яркостный бит не влияет.

Отметим также, что в начале 1991 г. изменена схема подключения цветного монитора (поменялись местами выходы красного и зеленого цветов).

Таблица 1

	Адрес
	Зона, управляемая соответствующими ячейками

	2400
2470
4700
6750
	Верхняя служебная строка
Собственно сам экран
Меню установок
Нижняя служебная строка


Приведем программу для ПП, выполняющую следующее: экран выключается, на нем (невидимо) рисуется голубой квадрат; экран включается (квадрат появляется); затем цвет квадрата начинает меняться от голубого до белого.

      MOV     PC,R0

      ADD     #SOOB-.,R0      ;Вычислить адрес сообщения
      MOV     R0,AMT
      MOV     PC,R0

      ADD     #SOOB1-.,R0     ;To же
      MOV     R0,AMT1
      MOV     #135230,@#2400  ;Разрешить вывод на экран всех цветов
      MOV     #177334,@#2402  ;для верхней служебной строки
      MOV     #135230,@#6750  ;То же сделать и для нижней
      MOV     #177334,@#6752

      MOV     #1,@#22614      ;Цвет символов в верхней строке – голубой
      MOV     #3,@#23106      ;Цвет символов в нижней строке – бирюзовый
      MOV     #63146,@#22620  ;Цвет знакоместа в верхней строке – желтый
      MOV     #52525,@#22622
      MOV     #52525,@#23112  ;Цвет знакоместа в нижней строке – фиолетовый
      MOV     #42104,@#23114
      MOV     #67,@#2372      ;Верхняя строка в формате 4 (10X24)
      MOV     #47,@#6742      ;Нижняя строка в формате 3 (20Х24)
      ЕМТ     52              ;Вывести сообщение в верхнюю строку
АМТ:  .WORD   0               ;Здесь находится адрес сообщения
      ЕМТ     56              ;Вывести сообщение в нижнюю строку
АМТ1: .WORD   0               ;Здесь находится адрес второго сообщения
      RTS     PC              ;Возврат в ЦП

SOOB:  .ASCIZ  <40.>/"ТЕХНОКОМ"/
SOOB1: .ASCIZ  <40.>"    ПРЕДСТАВЛЯЕТ    "
      .EVEN

При выводе информации в служебные строки можно менять формат и цвета символа и знакоместа, причем цвета меняются двумя способами. Первый — непосредственное изменение цветов символа и знакоместа, второй — через изменение палитры цветов вывода.

Пример программы с изменением цветов символа и знакоместа.

      MOV     PC,R0           ;Вычислить адрес сообщения
      ADD     #SOOB-.,R0
      MOV     R0,AMT

      MOV     PC,R0           ;То же

      ADD     #SOOB1-.,R0
      MOV     R0,AMT1
      EMT     52              ;Вывести "ТЕХНОКОМ" в верхнюю строку
АМТ:  .WORD   0               ;Здесь адрес сообщения

      ЕМТ     56              ;Вывести "ПРЕДСТАВЛЯЕТ" в нижнюю строку
АМТ1: .WORD   0               ;Здесь адрес второго сообщения
      MOV     #67,@#2372      ;Формат верхней строки - 4 (10X24)
      MOV     #47,@#6742      ;Формат нижней строки - 3 (20X24)
      MOV     #135236,@#2400  ;Для верхней строки представляем желтый цвет
      MOV     #174734,@#2402  ;голубым, а черный желтым

      MOV     #135235,@#6750  ;Для нижней строки представляем желтый цвет
      MOV     #175734,@#6752  ;бирюзовым, а черный фиолетовым

      RTS     PC              ;Возврат

SOOB:   .ASCIZ  <40.>/"ТЕХНОКОМ"/
SOOB1:  .ASCIZ  <40.>"    ПРЕДСТАВЛЯЕТ    "
      .EVEN

Ячейки с адресами 2400, 2402 (6750, 6752) отвечают за палитру вывода для верхней (нижней) служебной строки. Их формат был описан выше.

В ячейках 22614, 23106 хранится цвет символа в формате (0 — черный, 1 — голубой, 2 — зеленый, ... , 7 — белый), а в ячейках 23112, 23114 — цвет знакоместа в формате регистра кода цвета фона (177020, 177022) (этот формат будет рассмотрен ниже). В исходном состоянии (после включения питания) голубой план для вывода в служебные строки заблокирован.

Пример программы с изменением палитры вывода.

           4    3    2    1   - номер точки в октете
177020:  0000!0000!0000!0000

          RGB  RGB  RGB  RGB  - красный, зеленый, голубой планы

           8    7    6    5     Старший бит
177024:  0000!0000!0000!0000    каждой тетрады

          RGB  RGB  RGB  RGB    не используется

Требования при выводе текстовой информации в служебные строки таковы: первый байт любого сообщения есть байт позиционирования, а сам текст сообщения должен быть восьмибитным, так как коды <РУС> и <ЛАТ> (16(8) и 17(8) при печати не отрабатываются.

Теперь немного о выводе графической информации из ПП. Есть целая группа регистров, существенно облегчающая этот процесс:

177016 (запись/чтение) — регистр кода цвета точки. Формат — число от 0 до 7 (цвет точки);

177024 (запись) — регистр октета точки. Формат — восьмеричное число от 0 до 377 (восемь битов — восемь точек на экране; бит содержит 1 — точка зажжена, 0 — погашена);

177020 — регистр кода цвета фона (планы 0—1), 177022 — регистр кода цвета фона (планы 1—2). Оба регистра представляют собой единое целое; формат приведен на рисунке. Выведем на экран октет, в котором первая точка — черная, вторая — голубая; третья — зеленая, ..., восьмая — белая. Для этого нужно записать в 177020 число 0011 0010 0001 0000, а в 177022 — 0111 0110 0101 0100.

      MOV     #^B0011001000010000,@#177020 ;Загрузить регистр кода цвета
      MOV     #^B0111011001010100,@#177022 ;фона по принципу 1-я точка черная , восьмая – белая
      MOV     #100000,@#177010             ;Загрузить адрес планов в регистр адреса планов.
      MOV     #<264.*80.>,R1               ;В R1 общее число байтов экранной зоны
1$:   CLR     @#177024                     ;"Чистим" текущий октет
      INC     @#177010                     ;Увеличим на 1 адрес в планах

      SOB     R1,1$                        ;Bce это в цикле

;НА ЭКРАНЕ ВИДНЫ ЦВЕТНЫЕ ПОЛОСКИ
      MOV     #^B0100010101100111,@#177020 ;Загрузить другие значения
      MOV     #^B0000000100100011,@#177022 ;цвета фона
      MOV     PC,R2

      ADD     #SPRITE-.,R2                 ;Адрес спрайта в R2
      MOV     #100000,@#177010             ;Адрес в планах - 100000
      MOV     #7,@#177016                  ;Выводить будем белым
      MOV     #11,R0                       ;Размеры спрайта (Y - 11, X - 21)
3$:   MOV     #21,R1

;ЭТУ ЧАСТЬ ВЫДЕЛИМ

2$:   MOVB    (R2)+,@#177024     ;Байты, из которых состоит спрайт, по порядку на экран
;КОНЕЦ ВЫДЕЛЕННОЙ ЧАСТИ
      INC     @#177010           ;Увеличим адрес в планах на 1
      SOB     R1,2$              ;Следующий байт на экран

      ADD     #<80.-21>,@#177010 ;Отступить для вывода следующей строки
      SOB    R0,3$               ;спрайта.

      RTS     PC                 ;Возврат

;НИЖЕ ДАН САМ СПРАЙТ
SPRITE: .Byte 377,7,377,37,36,36,74,170,340,77,360,340,201,377,300,3,17
        .Byte 61,6,6,30,30,6,30,60,60,140,140,140,300,200,201,3,7
        .Byte 60,0,6,0,60,3,30,60,30,300,140,60,140,0,203,207,7
        .Byte 60,0,6,0,340,1,30,60,30,300,140,30,140,0,203,315,6
        .Byte 60,0,376,1,300,0,370,77,30,300,340,17,140,0,203,171,6
        .Byte 60,0,6,0,340,1,30,60,30,300,140,30,140,0,203,61,6
        .Byte 60,0,6,0,60,3,30,60,30,300,140,60,140,0,203,1,6
        .Byte 60,0,6,30,30,6,30,60,60,140,140,140,300,200,201,1,6
        .Byte 170,0,377,37,36,36,74,170,340,77,360,340,201,377,300,3,217
      .Even
ENDPRG:                          ;Конец программы в ПП

177026 — регистр маски планов. Формат — число от 0 до 7  (один или несколько планов, в которые нужно запретить запись информации). Регистр не блокирует запись в зеленый и красный планы со стороны ЦП.

Рассмотрим использование этих регистров. Приведенная ниже программа сначала заполняет экран цветными вертикальными полосами толщиной в одну точку, затем выводит в левом верхнем углу экрана спрайт.

2$:  MOVB    (R2)+,R3

     MOV     R3,@#177024

Вывод информации на экран может быть двух видов — с прорисовкой новой информации по старому фону и со стиранием старой информации. Чтобы рисовать по старому фону, нужно к регистру 177024 адресовать байтно (MOVB), для стирания — словно (MOV). В приведенном выше примере прорисовка со стиранием обеспечивается заменой выделенной части программы на следующую:

     BIS     #6,@#177710        ;Тактовая частота таймера 16 мкс
     BIC     #17400,@#177716    ;Очистить разряды с 8 по 12
     MOV     #200,R0            ;Установить в R0 единственный разряд – 7
1$:  XOR     R0,@#177716        ;Если он есть в 177716 ,то снять его и наоборот
     MOV #<от этого числа зависит высота звyкa>,R1

     SOB     R1,.               ;Задержка именно на этом числе

     BR      1$                 ;Бесконечный цикл

Как видно, с помощью этой группы регистров можно существенно повысить скорость вывода графики, однако иногда и этой скорости не хватает. Можно еще в несколько раз ее увеличить, используя нерегистровый (прямой) доступ к видеоОЗУ. Но об этом в следующий раз.

Теперь немного о музыке. Звук из динамика УКНЦ можно получить при помощи регистров 177710 и 177716. В регистре 177716 разряды с 7 по 12 позволяют получить звуки с частотой от 60 Гц до 64 кГц при частоте программируемого таймера 2, 4, 8 и 16 мкс. В табл. 2 даны соответствия комбинаций разрядов 8—16 регистра 177716, а в табл. 3 —разрядов 1, 2 регистра 177710.

Частоты звука приведены для тактовой частоты таймера 16 мкс. Если же тактовая частота равна 8 мкс, то значения частот в правой колонке увеличатся в 2 раза (соответственно для 4 мкс в 4 раза, а для 2 мкс в 8 раз).

Если разряды 8—12 содержат 0, то частота звука будет зависеть от того, с какой скоростью разряд 7 переключается из 0 в 1. Способ извлечения звука прост. Меняя значение, загружаемое для задержки, мы управляем высотой звука. Программа, приведенная ниже, работает именно по этому принципу.
Таблица 2

	Разряды регистра 177716
	Частота звукового сигнала

	7
	12
	11
	10
	9
	8
	

	Программно формируется
	0
	0
	0
	0
	0
	Соответствует программе

	1
	0

0

0

0

1
	0

0

0

1

0
	0

0

1

0

0
	0

1

0

0

0
	1

0

0

0

0
	60 Гц

250 Гц

500 Гц

1 кГц

8 кГц

	0
	Звук отключен
	

	1
	Комбинация логических значений
	Сборка сетки частот по «И»

	Программно формируется
	*
	*
	*
	*
	*
	Сборка сетки частот по «И» промодулирована программой


1-я октава

      До - 130,8

      До-диез, ре-бемоль - 138,6

      Ре - 146,8

      Ре-диез, ми-бемоль — 155,6

      Ми - 164,8

      Фа - 174,6

      Фа-диез, соль-бемоль - 185,0

      Соль - 196,0

      Соль-диез, ля-бемоль - 207,7

      Ля - 220,0

      Ля-диез, си-бемоль - 233,1

      Си - 246,9

      Среднее до - 261,7
2-я октава

      До-диез, ре-бемоль - 277,2

      Ре - 293,7

      Ре-диез, ми-бемоль - 311,1

      Ми - 329,6

      Фа - 349,2

      Фа-диез, соль-бемоль - 370,0

      Соль - 392,0

      Соль-диез, ля-бемоль - 415,3

      Ля - 440,0

      Ля-диез, си-бемоль - 466,2

      Си - 493,9

      Среднее до - 523,3

Этот способ имеет существенный недостаток. На разных машинах высота звука будет «плавать», так как быстродействие ПП меняется от машины к машине из-за разброса параметров элементной базы. Устранить данный недостаток можно, «собирая» нужные частоты из набора стандартных. Сборка ведется по «И» (одна частота модулирует собой другую), что позволяет, управляя частотой программируемого таймера и комбинируя значения разрядов 8—12 регистра 177716, добиваться исполнения несложных мелодий.

Таблица 3

	Разряды регистра 177710
	Период тактовой частоты

	2
	1
	

	0

0

1

1
	0

1

0

1
	2 мкс

4 мкс

8 мкс

16 мкс


Данный способ дает стабильный звук, но его недостаток очевиден: крайне трудно получить требуемые ноты. Гораздо лучше использовать программируемый таймер в «чистом» виде в режиме прерываний. Достоинства такого способа: чистый стабильный звук, подбор частоты до единиц герц и легкость практической реализации. Действительно, раз можно плавно менять высоту звука и его длительность (это реализуется на другом таймере — сетевом), то уже можно исполнять довольно сложные мелодии. Для облегчения задачи приведем значения частот звучания нот двух октав в герцах.

       MOV     PC,R0           ;Вычислить адрес программы обработки прерываний
       ADD     #POP-.,R0       ;от программируемого таймера
       MOV     R0,@#304        ;Загрузить этот адрес в вектор прерывания
       MOV     PC,R0           ;Вычислить адрес программы обработки прерываний
       ADD     #POPT-.,R0      ;от сетевого таймера

       MOV     R0,@#100        ;Загрузить этот адрес в вектор прерывания
       MOV     PC,R0           ;Вычислить начальный адрес массива с музыкой
       ADD     #MUZON-.,R0
       MOV     R0,ADRESS       ;Загрузить этот адрес в ячейку ADRESS
       MOV     R0,EADRES       ;Вычислить и загрузить в ячейку EADRES
       ADD     #<EP-MUZ0N>,EADRES ;конечный адрес массива с музыкой
       BIS     #107,@#177710   ;Установить период тактовой частоты программируемого
       RTS     PC              ;таймера 16 мкс и разрешить ;прерывания от него
;ПРОГРАММА ОБРАБОТКИ ПРЕРЫВАНИЙ ОТ СЕТЕВОГО ТАЙМЕРА
РОРТ: .ENABLE LSB

       DEC     (РС)+           ;Уменьшить на 1 содержимое ячейки COUNT
COUNT:  .WORD   50.            ;50 тиков - это 1 секунда (длительность ноты)
       BNE     1$              ;Если секунда не истекла, то на выход
       MOV     @ADRESS,TON+2   ;Сыграть очередную ноту из массива
       ADD     #2,ADRESS       ;Перейти на следующую ноту

       CMP     EADRES,ADRESS   ;А не конец ли массива?
       BNE     2$
       SUB     #<EP-MUZ0N>,ADRESS ;Да. Вернемся к началу

2$:  MOV     #50.,COUNT        ;1 секунда в COUNT
1$:  JMP     @#174612          ;Выход из прерывания

;ПРОГРАММА ОБРАБОТКИ ПРЕРЫВАНИЙ ОТ ПРОГРАММИРУЕМОГО ТАЙМЕРА
РОР: .ENABLE LSB;

       TST     @#177714        ;Очистить регистр текущего значения пр. тайм.
       MOV     R0,-(SP)        ;Сохранить в стеке R0
       MOV     #200,R0         ;Загрузить число 200 в R0
       XOR     R0,@#177716     ;7 разряд если есть - снять, и наоборот
       MOV     (SP)+,R0        ;Восстановить R0

Пример программы, иллюстрирующей применение программируемого таймера:

TON: MOV     #0,@#177712    ;Загрузить текущее значение ячейки TON+2 в буферный регистр программируемого таймера
       RTI                  ;Возврат из прерывания

ADRESS: .WORD   0           ;Ячейка хранения начального адреса массива

EADRES: .WORD   0           ;Ячейка хранения конечного адреса массива
;ДАЛЕЕ СЛЕДУЕТ САМ МАССИВ С МУЗЫКОЙ

MUZON:  .WORD   478.    ;До первой октавы
        .WORD   426.    ;Ре первой октавы
        .WORD   379.    ;Ми первой октавы
        .WORD   358.    ;Фа первой октавы
        .WORD   319.    ;Соль первой октавы
        .WORD   284.    ;Ля первой октавы
        .WORD   253.    ;Си первой октавы
        .WORD   239.    ;Срединное До
        .WORD   213.    ;Ре второй октавы
        .WORD   190.    ;Ми второй октавы
        .WORD   179.    ;Фа второй октавы
        .WORD   159.    ;Соль второй октавы
        .WORD   142.    ;Ля второй октавы
        .WORD   127.    ;Си второй октавы
        .WORD   119.    ;До второй октавы

ЕР:                         ;Конец программы в ПП

Эта программа проигрывает указанные ноты по порядку в цикле. Значения чисел, соответствующие реальным звуковым частотам, рассчитываются по формуле A/F, где A соответствует тактовой частоте программируемого таймера (для периода 16 мкс A= 62500, для 8 мкс A= 125000, для 4 мкс A=250000, для 2 мкс A= 500000 (все значения десятичные), a F — частота, которую требуется получить. Обратите внимание на то, что буферный регистр программируемого таймера (177712) 12-, а не 16-разрядный! Об этом следует помнить при загрузке данных в регистр.

